D02QGF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

D02QGF is a reverse communication routine for integrating a non-stiff system of first-order ordinary differential equations using a variable-order variable-step Adams method. A root-finding facility is provided.

2 Specification

```
 SUBROUTINE D02QGF(NEQF, T, Y, TOUT, NEQG, ROOT, IREVCM, TRVCM,
1                  YRVCM, YPRVCM, GRVCM, KGRVCM, RWORK, LRWORK,
2                  IWORK, LIWORK, IFAIL)
 INTEGER           NEQF, NEQG, IREVCM, YRVCM, YPRVCM, KGRVCM,
1                  LRWORK, IWORK(LIWORK), LIWORK, IFAIL
 real              T, Y(NEQF), TOUT, TRVCM, GRVCM, RWORK(LRWORK)
 LOGICAL           ROOT
```

3 Description

Given the initial values $x, y_1, y_2, \ldots, y_{\mathrm{NEQF}}$ the routine integrates a non-stiff system of first-order differential equations of the type, $y_i' = f_i(x, y_1, y_2, \ldots, y_{\mathrm{NEQF}})$, for $i = 1, 2, \ldots, \mathrm{NEQF}$, from $x = \mathrm{T}$ to $x = \mathrm{TOUT}$ using a variable-order variable-step Adams method. The user defines the system by reverse communication, evaluating f_i in terms of x and $y_1, y_2, \ldots, y_{\mathrm{NEQF}}$, and $y_1, y_2, \ldots, y_{\mathrm{NEQF}}$ are supplied at $x = \mathrm{T}$ by D02QGF. The routine is capable of finding roots (values of x) of prescribed event functions of the form

$$g_j(x, y, y') = 0, \quad j = 1, 2, \ldots, \mathrm{NEQG}.$$

Each g_j is considered to be independent of the others so that roots are sought of each g_j individually. The root reported by the routine will be the first root encountered by any g_j. Two techniques for determining the presence of a root in an integration step are available: the sophisticated method described in Watts [3] and a simplified method whereby sign changes in each g_j are looked for at the ends of each integration step. The user also defines each g_j by reverse communication. In one-step mode the routine returns an approximation to the solution at each integration point. In interval mode this value is returned at the end of the integration range. If a root is detected this approximation is given at the root. The user selects the mode of operation, the error control, the root-finding technique and various optional inputs by a prior call of the setup routine D02QWF.

For a description of the practical implementation of an Adams formula see Shampine and Gordon [1].

4 References

[1] Shampine L F and Gordon M K (1975) *Computer Solution of Ordinary Differential Equations – The Initial Value Problem* W H Freeman & Co., San Francisco

[2] Shampine L F and Watts H A (1979) DEPAC – design of a user oriented package of ODE solvers *Report SAND79–2374* Sandia National Laboratory

[3] Watts H A (1985) RDEAM – An Adams ODE code with root solving capability *Report SAND85–1595* Sandia National Laboratory

5 Parameters

Note: this routine uses **reverse communication.** Its use involves an initial entry, intermediate exits and re-entries, and a final exit, as indicated by the **parameter IREVCM**. Between intermediate exits and re-entries, **all parameters other than GRVCM and RWORK must remain unchanged**.

1: NEQF — INTEGER *Input*

On initial entry: the number of first-order ordinary differential equations to be solved by D02QGF. It must contain the same value as the parameter NEQF used in the prior call to D02QWF.

Constraint: NEQF ≥ 1.

2: T — ***real*** *Input/Output*

On initial entry: that is after a call to D02QWF with STATEF = 'S', T must be set to the initial value of the independent variable x.

On final exit: the value of x at which y has been computed. This may be an intermediate output point, a root, TOUT or a point at which an error has occurred. If the integration is to be continued, possibly with a new value for TOUT, T must not be changed.

3: Y(NEQF) — ***real*** array *Input/Output*

On initial entry: the initial values of the solution $y_1, y_2, \ldots, y_{\mathrm{NEQF}}$.

On final exit: the computed values of the solution at the exit value of T. If the integration is to be continued, possibly with a new value for TOUT, these values must not be changed.

4: TOUT — ***real*** *Input*

On initial entry: the next value of x at which a computed solution is required. For the initial T, the input value of TOUT is used to determine the direction of integration. Integration is permitted in either direction. If TOUT = T on exit, TOUT must be reset beyond T **in the direction of integration,**before any continuation call.

5: NEQG — INTEGER *Input*

On initial entry: the number of event functions which the user is defining for root-finding. If root-finding is not required the value for NEQG must be ≤ 0. Otherwise it must be the same value as the parameter NEQG used in the prior call to D02QWF.

6: ROOT — LOGICAL *Output*

On final exit: if root-finding was required (NEQG > 0 on entry), then ROOT specifies whether or not the output value of the parameter T is a root of one of the event functions. If ROOT = .FALSE., then no root was detected, whereas ROOT = .TRUE. indicates a root and the user should make a call to D02QYF for further information.

If root-finding was not required (NEQG = 0 on entry), then ROOT = .FALSE..

7: IREVCM — INTEGER *Input/Output*

On initial entry: IREVCM must have the value 0.

On intermediate exit: IREVCM specifies what action the user must take before re-entering D02QGF **with** IREVCM **unchanged**. The possible values of IREVCM on exit from D02QGF are 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12 which should be interpreted as follows:

IREVCM = 1, 2, 3, 4, 5, 6 or 7

indicates that the user must supply $y' = f(x, y)$, where x is given by TRVCM and y_i is returned in Y(i), for $i = 1, 2, \ldots,$ NEQF when YRVCM = 0 and RWORK(YRVCM + $i - 1$), for $i = 1, 2, \ldots,$ NEQF when YRVCM $\neq 0$. y'_i should be placed in location RWORK(YPRVCM+$i-1$), for $i = 1, 2, \ldots,$ NEQF.

IREVCM = 8

indicates that the current step was not successful due to error test failure. The only information supplied to the user on this return is the current value of the independent variable T, as given by TRVCM. No values must be changed before re-entering D02QGF. This facility enables the user to determine the number of unsuccessful steps.

IREVCM = 9, 10, 11, or 12

indicates that the user must supply $g_k(x, y, y')$, where k is given by KGRVCM, x is given by TRVCM, y_i is given by Y(i) and y'_i is given by RWORK(YPRVCM $- 1 +$ i). The result g_k should be placed in the variable GRVCM.

On final exit: IREVCM has the value 0, which indicates that an output point or root has been reached or an error has occurred (see IFAIL).

8: TRVCM — ***real*** *Output*

On intermediate exit: the current value of the independent variable.

9: YRVCM — INTEGER *Output*

On intermediate exit: with IREVCM = 1, 2, 3, 4, 5, 6, 7, 9, 10, 11 or 12, YRVCM specifies the locations of the dependent variables y for use in evaluating the differential system or the event functions. If YRVCM = 0 then y_i is given by Y(i), for $i = 1, 2, \ldots,$ NEQF. If YRVCM $\neq 0$ then y_i is given by RWORK(YRVCM $+ i - 1$), for $i = 1, 2, \ldots,$ NEQF.

10: YPRVCM — INTEGER *Output*

On intermediate exit: with IREVCM = 1, 2, 3, 4, 5, 6, or 7, YPRVCM specifies the positions in RWORK at which the user should place the derivatives y'. y'_i should be placed in location RWORK(YPRVCM $+ i - 1$), for $i = 1, 2, \ldots,$ NEQF.

With IREVCM = 9, 10, 11 or 12, YPRVCM specifies the locations of the derivatives y' for use in evaluating the event functions. y'_i is given by RWORK(YPRVCM $+ i - 1$), for $i = 1, 2, \ldots,$ NEQF. YPRVCM must not be changed before re-entering D02QGF.

11: GRVCM — ***real*** *Input*

On intermediate re-entry: with IREVCM = 9, 10, 11 or 12, GRVCM must contain the value of $g_k(x, y, y')$, where k is given by KGRVCM.

12: KGRVCM — INTEGER *Output*

On intermediate exit: with IREVCM = 9, 10, 11 or 12, KGRVCM specifies which event function $g_k(x, y, y')$ the user must evaluate.

13: RWORK(LRWORK) — ***real*** array *Workspace*

This **must** be the same parameter RWORK as supplied to D02QWF. It is used to pass information from D02QWF to D02QGF, and from D02QGF to the D02QXF, D02QYF and D02QZF. Therefore the contents of this array **must not** be changed before the call to D02QGF or calling any of the routines D02QXF, D02QYF and D02QZF.

14: LRWORK — INTEGER *Input*

On initial entry: the dimension of the array RWORK as declared in the (sub)program from which D02QGF is called.

This must be the same parameter LRWORK as supplied to D02QWF.

15: IWORK(LIWORK) — INTEGER array *Workspace*

This **must** be the same parameter IWORK as supplied to D02QWF. It is used to pass information from D02QWF to D02QGF, and from D02QGF to D02QXF, D02QYF and D02QZF. Therefore the contents of this array **must not** be changed before the call to D02QGF or calling any of the routines D02QXF, D02QYF and D02QZF.

16: LIWORK — INTEGER *Input*

On initial entry: the dimension of the array IWORK as declared in the (sub)program from which D02QGF is called.

This must be the same parameter LIWORK as supplied to D02QWF.

17: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL ≠ 0 on exit, users are recommended to set IFAIL to −1 before entry. **It is then essential to test the value of IFAIL on exit**.

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL = 1

On entry, the integrator detected an illegal input, or D02QWF has not been called prior to the call to the integrator. If on entry IFAIL = 0 or −1, the form of the error will be detailed on the current error message unit (as defined by X04AAF).

This error may be caused by overwriting elements of RWORK and IWORK.

IFAIL = 2

The maximum number of steps has been attempted (at a cost of about 2 derivative evaluations per step). (See parameter MAXSTP in D02QWF.) If integration is to be continued then the user need only reset IFAIL and call the routine again and a further MAXSTP steps will be attempted.

IFAIL = 3

The step size needed to satisfy the error requirements is too small for the ***machine precision*** being used. (See parameter TOLFAC in D02QXF.)

IFAIL = 4

Some error weight w_i became zero during the integration (see parameters VECTOL, RTOL and ATOL in D02QWF.) Pure relative error control (ATOL = 0.0) was requested on a variable (the ith) which has now become zero. (See parameter BADCMP in D02QXF.) The integration was successful as far as T.

IFAIL = 5

The problem appears to be stiff (see the Chapter Introduction for a discussion of the term 'stiff'). Although it is inefficient to use this integrator to solve stiff problems, integration may be continued by resetting IFAIL and calling the routine again.

IFAIL = 6

A change in sign of an event function has been detected but the root-finding process appears to have converged to a singular point T rather than a root. Integration may be continued by resetting IFAIL and calling the routine again.

IFAIL = 7

The code has detected two successive error exits at the current value of T and cannot proceed. Check all input variables.

7 Accuracy

The accuracy of integration is determined by the parameters VECTOL, RTOL and ATOL in a prior call to D02QWF. Note that only the local error at each step is controlled by these parameters. The error estimates obtained are not strict bounds but are usually reliable over one step. Over a number of steps the overall error may accumulate in various ways, depending on the property of the differential equation system. The code is designed so that a reduction in the tolerances should lead to an approximately proportional reduction in the error. The user is strongly recommended to call D02QGF with more than one set of tolerances and to compare the results obtained to estimate their accuracy.

The accuracy obtained depends on the type of error test used. If the solution oscillates around zero a relative error test should be avoided, whereas if the solution is exponentially increasing an absolute error test should not be used. If different accuracies are required for different components of the solution then a component-wise error test should be used. For a description of the error test see the specifications of the parameters VECTOL, ATOL and RTOL in the routine document for D02QWF.

The accuracy of any roots located will depend on the accuracy of integration and may also be restricted by the numerical properties of $g(x, y, y')$. When evaluating g the user should try to write the code so that unnecessary cancellation errors will be avoided.

8 Further Comments

If the routine fails with IFAIL = 3 then the combination of ATOL and RTOL may be so small that a solution cannot be obtained, in which case the routine should be called again with larger values for RTOL and/or ATOL. If the accuracy requested is really needed then the user should consider whether there is a more fundamental difficulty. For example:

(a) in the region of a singularity the solution components will usually be of a large magnitude. The routine could be used in one-step mode to monitor the size of the solution with the aim of trapping the solution before the singularity. In any case numerical integration cannot be continued through a singularity, and analytical treatment may be necessary;
(b) for 'stiff' equations, where the solution contains rapidly decaying components, the routine will require a very small step size to preserve stability. This will usually be exhibited by excessive computing time and sometimes an error exit with IFAIL = 3, but usually an error exit with IFAIL = 2 or 5. The Adams methods are not efficient in such cases and the user should consider using a routine from the subchapter D02M–D02N. A high proportion of failed steps (see parameter NFAIL in D02QXF) may indicate stiffness but there may be other reasons for this phenomenon.

D02QGF can be used for producing results at short intervals (for example, for graph plotting); the user should set CRIT = .TRUE. and TCRIT to the last output point required in a prior call to D02QWF and then set TOUT appropriately for each output point in turn in the call to D02QGF.

9 Example

We solve the following system (for a projectile)

$$\begin{aligned} y' &= \tan\phi \\ v' &= \frac{-0.032\tan\phi}{v} - \frac{0.02v}{\cos\phi} \\ \phi' &= \frac{-0.032}{v^2} \end{aligned}$$

over an interval $[0.0, 10.0]$ starting with values $y = 0.5$, $v = 0.5$ and $\phi = \pi/5$ using scalar error control (VECTOL = .FALSE.) until the first point where $y = 0.0$ is encountered.

Also, we use D02QGF to produce output at intervals of 2.0.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     D02QGF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NEQF, NEQG, LATOL, LRTOL, LRWORK, LIWORK
      PARAMETER        (NEQF=3,NEQG=1,LATOL=1,LRTOL=1,
     +                 LRWORK=23+23*NEQF+14*NEQG,LIWORK=21+4*NEQG)
      real             TSTART, HMAX
      PARAMETER        (TSTART=0.0e0,HMAX=2.0e0)
*     .. Local Scalars ..
      real             GRVCM, PI, T, TCRIT, TINC, TOUT, TRVCM
      INTEGER          I, IFAIL, IREVCM, J, KGRVCM, MAXSTP, YPRVCM,
     +                 YRVCM
      LOGICAL          ALTERG, CRIT, ONESTP, ROOT, SOPHST, VECTOL
      CHARACTER*1      STATEF
*     .. Local Arrays ..
      real             ATOL(LATOL), RTOL(LRTOL), RWORK(LRWORK), Y(NEQF)
      INTEGER          IWORK(LIWORK)
*     .. External Functions ..
      real             X01AAF
      EXTERNAL         X01AAF
*     .. External Subroutines ..
      EXTERNAL         D02QGF, D02QWF
*     .. Intrinsic Functions ..
      INTRINSIC        COS, real, TAN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D02QGF Example Program Results'
      TCRIT = 10.0e0
      STATEF = 'S'
      VECTOL = .FALSE.
      RTOL(1) = 1.0e-4
      ATOL(1) = 1.0e-7
      ONESTP = .FALSE.
      SOPHST = .TRUE.
      CRIT = .TRUE.
      TINC = 2.0e0
      MAXSTP = 500
      PI = X01AAF(0.0e0)
      T = TSTART
      Y(1) = 0.5e0
      Y(2) = 0.5e0
      Y(3) = 0.2e0*PI
      WRITE (NOUT,*)
      WRITE (NOUT,*) '  T         Y(1)     Y(2)     Y(3)'
      WRITE (NOUT,99999) T, (Y(I),I=1,NEQF)
      IFAIL = 0
*
      CALL D02QWF(STATEF,NEQF,VECTOL,ATOL,LATOL,RTOL,LRTOL,ONESTP,CRIT,
     +            TCRIT,HMAX,MAXSTP,NEQG,ALTERG,SOPHST,RWORK,LRWORK,
     +            IWORK,LIWORK,IFAIL)
*
      J = 1
      TOUT = real(J)*TINC
```

```
      IREVCM = 0
*
   20 IFAIL = -1
*
      CALL D02QGF(NEQF,T,Y,TOUT,NEQG,ROOT,IREVCM,TRVCM,YRVCM,YPRVCM,
     +            GRVCM,KGRVCM,RWORK,LRWORK,IWORK,LIWORK,IFAIL)
*
      IF (IREVCM.GT.0) THEN
         IF (IREVCM.LT.8) THEN
            IF (YRVCM.EQ.0) THEN
               RWORK(YPRVCM) = TAN(Y(3))
               RWORK(YPRVCM+1) = -0.032e0*TAN(Y(3))/Y(2) - 0.02e0*Y(2)
     +                           /COS(Y(3))
               RWORK(YPRVCM+2) = -0.032e0/Y(2)**2
            ELSE
               RWORK(YPRVCM) = TAN(RWORK(YRVCM+2))
               RWORK(YPRVCM+1) = -0.032e0*TAN(RWORK(YRVCM+2))
     +                           /RWORK(YRVCM+1) - 0.02e0*RWORK(YRVCM+1)
     +                           /COS(RWORK(YRVCM+2))
               RWORK(YPRVCM+2) = -0.032e0/RWORK(YRVCM+1)**2
            END IF
         ELSE IF (IREVCM.GT.8) THEN
            GRVCM = Y(1)
         END IF
         GO TO 20
      ELSE IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,99999) T, (Y(I),I=1,NEQF)
         IF (T.EQ.TOUT .AND. J.LT.5) THEN
            J = J + 1
            TOUT = real(J)*TINC
            GO TO 20
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,F6.4,3X,3(F7.4,2X))
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D02QGF Example Program Results

   T         Y(1)     Y(2)     Y(3)
 0.0000    0.5000   0.5000   0.6283
 2.0000    1.5493   0.4055   0.3066
 4.0000    1.7423   0.3743  -0.1289
 6.0000    1.0055   0.4173  -0.5507
 7.2883    0.0000   0.4749  -0.7601
```